ATTACHMENT C
DOCUMENTATION OF ULTRACONE.xls

Ultracone uses simple calculations of the critical pressure based on properties of the lowermost USDW and the injection zone to calculate the pressure increase needed to cause liquid from the injection zone to move upward through an open conduit and into the lowermost USDW. It also calculates the radius within which pressures will exceed that level.

DATA VALUES WITHIN IN THE PRINTABLE AREA OF THE ZEI WORKSHEET

1. **Data that must be entered directly into the ZEI worksheet (green cells):**

   E8 Analyst
   C13 Duration of Additional Injection, mos
   C28 Feet to Potential Conduit

2. **Data that may be either automatically returned or manually entered (yellow cells).** If no reference value is entered on the "Transfer" sheet, or a different value than the one calculated is more appropriate, some of these cells may be overwritten.

   C11 Injection Rate, gpm (must show up as a negative number)
   D11 Minimum Specific Gravity (of injectate)
   E13 Cumulative Injected Volume – Should be entered directly into Cell E13. Otherwise, it is (inaccurately but conservatively) calculated based on nominal maximum rate of injection and duration of past injection. [DAYS360(E11,D8)*1440*Transfer!B102]

3. **Data automatically returned:**

   B13 Radius of Injection Well, ft. – Converts standard notation of diameter in inches to radius in feet. [=IF(Transfer!B93>0,Transfer!B93,Transfer!B75)/24]
   E11 Date of First Injection - Uses today's date if no date has been entered in the data transfer sheet (not required, used to calculate plume radius) [=IF(Transfer!B99>0,Transfer!B99,TODAY())]

4. **Data automatically transferred from the data transfer sheet to protected cells in the worksheet:**

   <u>For administrative information:</u>
   B4 Facility Name
   D4 Operator
   B6 Well Name
   D6 U.S. EPA Permit Number
   E6 Well Class
   B8 County
   C8 State

D8 Date - The date is entered automatically to ensure that printed versions are the most recent and for simple differentiation.

For well and operational information:
B11 Surface Elevation of Well, ft. (not required if information is all from the subject well)
D13 Viscosity of Injectate

For USDW information:
B16 Name of USDW (not required for calculation)
C16 Hydrostatic Pressure in USDW, psi
D16 Specific Gravity of Water in USDW
B18 Depth to Base of USDW, ft.
C18 Depth of Pressure Measurement, ft.

For injection zone information:
B21 Formation at Top of Injection Zone
C21 Porosity of Injection zone
D21 Specific Gravity of Liquid in the Injection zone
B23 Depth to Top of Injection Zone, ft.
C23 Permeability of Injection Zone, ft.
E23 Measured Pressure in Injection Zone, psi
B25 Effective Thickness of Injection Zone, ft.
C25 Compressibility of Injection Zone, $ft^3/ft^3/psi$
E25 Depth of Pressure Measurement, ft.

CALCULATIONS WITHIN IN THE PRINTABLE AREA OF THE ZEI WORKSHEET

Cell formulas are shown with row & column cell references only, then with cell names, if applicable.

Cumulative Injected Volume, gals.

Cell E13 =DAYS360(E11,D8)*1440*Transfer!B103

This is V_{past}, and may be entered directly if known (preferred) or calculated based on the maximum permitted injection rate and the duration of past injection.

Pressure at the base of USDW, psia

Cell D18 =C16+(B18-C18)*0.433*D16+14.7
=psiusdw+(dusdw-usdwpd)*0.433*usdwsg+14.7

The pressure at the base of the USDW is calculated using the equation:

$$p_{USDW} = p_{meas} + (d_{base} - d_{meas}) \times SG_{USDW} \times 0.433 psi/ft \qquad \text{Eq. 1}$$

where:
p_{USDW} = pressure at the base of the lowermost underground source of drinking water, in psia;
p_{meas} = a measured pressure reflecting hydrostatic pressure within the USDW, in psig;
d_{base}= the depth to the base of the lowermost USDW, in feet;
d_{meas} = the depth at which the pressure measurement was made,
SG_{USDW} = the specific gravity of the water in the USDW, and
0.433 psi/ft = the liquid pressure gradient for fresh water.

Translation: The pressure at the base of the USDW is equal to the measured pressure (psig) plus the height of the liquid column separating the depth of the measurement and the base of the USDW multiplied by the liquid pressure gradient of the liquid in the column plus the approximate influence of atmospheric pressure [14.7] in psi. Gauge pressure is used because very often pressure measurements are based on a liquid level at a depth which is affected by atmospheric pressure although it is recorded as zero psi.

Pressure at the top of the injection zone.

Cell E21:
```
=E23-D21*0.433*(E25-B23)
=iipsi-izsg*0.433*(iipsid-izt)
```

The pressure at the top of the injection zone or interval is calculated from a measurement of pressure representing the hydrostatic pressure in the injection zone. It is calculated using an equation similar to Eq. 1, but using data appropriate for the injection zone. Gauges which measure hydrostatic pressure usually include the effects of atmospheric pressure in their readings.

Viscosity of Connate Fluid, md

Cell D23
```
=IF('Transfer'B54>0, 'Transfer'B54,IF(I35>0,SUM(G36:G43),))
=IF(izvis>0,izvis,IF(I35>0,SUM(G36:G43),))
```

Translation: If a value for reservoir fluid viscosity was entered, return it; otherwise, if a reservoir temperature was entered, retrieve the viscosity value calculated from it; otherwise, leave blank.

Compressibility of connate fluid

Cell D25
```
=IF(Transfer!B51>0,1/(7.033*E21+0.05413*Transfer!B51-537*
    IF(E30=D21,Transfer!B59,70)+403.3*10^3),0)
=IF(cl>0,cl,IF(iztds>0,1/(7.033*E21+0.05413*iztds-537*
    IF(effsg=D21,iztemp,70)+403.3*10^3),0))
```

This calculation is based on Equation 2 on Page 24-13 of the Petroleum Engineering Handbook:

$$1/c_w \equiv m_1 \times p + m_2 \times C + m_3 \times T + m_4 \qquad \text{Eq. 2}$$

Where:

c_w = Compressibility of water,
p = Hydrostatic pressure,
C = Salinity, mg/l,
T = Temperature, degrees Fahrenheit,
m_1 = 7.033
m_2 = 0.05413
m_3 = -537
m_4 = 403.3 x 10^3

Translation: If a value has been entered for total dissolved solids divide 1 by the sums of constant m_1 times the average formation pressure, constant m_2 times the total dissolved solids concentration of the connate water, constant m_3 times the average temperature of the injection zone if the specific gravity of water in the injection zone was chosen as appropriate, otherwise 70 °F, and constant m_4, Otherwise (if no value was entered for dissolved solids), leave blank.

Critical pressure

Cell B28 =IF(E30>0,D18+(B23-B18)*0.433*E30-E21,"Choose Viscosity")
=if(effsg>0,D18+(izt-dusdw)*0.433*effsg-E21,"Choose Viscosity")

A critical pressure is computed if all of the cells requiring data are filled. An if-then statement requires that the appropriate specific gravity be specified in cell E30. The pressure is computed using the equation:

$$p_c = p_{USDW} + (d_{iz} - d_{base}) \times SG_{iz} \times 0.433 psi/ft - p_{iz} \qquad \text{Eq. 3}$$

Where:

p_c = the critical pressure addition needed to cause flow of liquid from the injection zone, through an open conduit, and into the lowermost USDW, in psi;
d_{iz} = the depth to the top of the injection zone, in feet;
SG_{iz} = the specific gravity of the liquid in the injection zone near the distance of the radius of endangering influence; and
p_{iz} = the hydrostatic pressure at the top of the injection zone, in psi.

Translation: If a value has been entered for specific gravity to use, calculate the critical pressure by adding to pressure at the base of the lowermost USDW the product of the pressure which would be exerted by a column of liquid reaching from the top of the injection zone to the base of the USDW minus the pressure already existing at the top of the injection zone.

The radius of the cone of endangering influence.

Cell D28 =IF($D$30>0,IF(B28>B13,SQRT($C$23*$H$9/($C$21*$D$30*($C$25+$D$25)*10^($B$28*$C$23*$B$25/(162.6*$G$9*$D$30)+3.23))),"INFINITE"),"Fill Blanks")
=IF(effvis>0,IF(cp>B13,SQRT(izhk*t/(izpor*effvis*(cfm+D25)*10^(-cp*izhk*izh/(162.6*q*effvis)+3.23))),"INFINITE"),"Fill blanks")

This is computed using the criteria in 40 CFR 146.6(a)(ii)(2) with the exponential integral method for calculating pressure in horizontal reservoirs. The equation is couched in reservoir engineering terms, more commonly used when dealing with deep reservoirs rather than hydrological terms, more commonly used when dealing with near surface reservoirs. A result will be calculated if the blanks are filled and if the radius of the ZEI is greater than 0.5 feet in diameter:

$$r_{ZEI} = \sqrt{\frac{k \times t}{\phi \times \mu \times (c_f + c_w) \times 10^x}} \qquad \text{Eq. 4}$$

Where:

- k = permeability, md
- t = time, hours
- ϕ = the porosity of the effective injection zone
- μ = viscosity of liquid dominating pressure buildup in the injection zone, cp
- c_f = compressibility of the formation, psi^{-1}
- c_w = compressibility of water psi^{-1}
- x = Equation 5

$$x = \frac{p_c \times k \times h}{162.6 \times q \times \mu} + 3.23 \qquad \text{Eq. 5}$$

Where:

- h = effective thickness of the injection zone/interval, ft
- q = injection rate, gpm

Translation: If a value has been entered for the specific gravity to be used; if the critical pressure is larger than an arbitrarily small number (greater than the radius of the well bore), the square root of: permeability multiplied by time of injection divided by the product of the porosity and viscosity and compressibility of the formation and the contained water times 10 raised to the x^th power (x defined as 3.23 plus the quotient of the product of -1, critical pressure, permeability and thickness, divided by the product of 162.6, injection rate and viscosity; if the critical pressure is not greater than an arbitrarily small number, then display "INFINITE". If no value has been entered for the specific gravity to be used, display "Choose SG".

Radius of the waste plume, some dispersion

Cell B30 IF(B25*C21>0,SQRT((E13+C11*C13*1440*30.44)*0.1337*3/(B25*C21*PI())),0)
=IF(izh*izpor>0,SQRT(totvol*0.1337*3/(izh*izpor*PI())),0)

This computed result is only for the purpose of getting a relative idea of the size of the plume. The distance beyond which there is essentially no migration of waste will actually be as though three times the volume of waste had been injected. The radius is a simple calculation of the radius of a cylinder of porous material with the volume determined by the total injected volume multiplied by three using equation 9:

$$r_{wp} = \sqrt{\frac{V_{cum} \times 0.1337 \times 3}{h \times \phi \times \pi}} \qquad \text{Eq. 6}$$

Where:

- r_{wp} = the radius of the waste plume, assuming no dispersion
- V_{cum} = the cumulative injection at the end of the well's projected life
- 0.1337 = factor to convert gallons to cubic feet
- 3 = factor to roughly account for dispersion

Maximum Injection Rate Label

Cell C29 = IF(C28>0,"Max Safe Injection Rate, gpm","")

Translation: If the distance to a conduit was entered in Cell C29, display "Max Safe Injection Rate, gpm; otherwise, display an invisible zero.

Distance beyond which injection pressure is below critical pressure

Cell C30 =IF(C28>0,42/1440*$C$23*$B$25*$B$28/(162.6*$D$30)/(LOG10($C$23*$H$9/($C$21*$D$30*($C$25+$D$25)*$C$28^2))-3.23),"")

Calculation of the radial distance at which injection pressure falls below critical pressure. The log approximation of the Ei method is adequate for the time periods typical of injection well lives. The source equation is from Earlougher, 1977, Equation 3.4.

$$p_{WF} = p_i - 162.6 \times \frac{qB\mu}{kh} \times \left(\log\left(\frac{kt}{\phi \mu c_t r_c^{\,2}} \right) - 3.23 \right) \qquad \text{Eq. 7}$$

Where:

- p_{wf} = flowing pressure, psi
- p_i = initial pressure, psi
- r_c = radial distance to potential conduit

Setting initial pressure equal to zero ($p_i = 0$), and solving for q, the resulting equation for the pressure change, which will be the critical pressure, p_c, is:

$$q = \frac{khp_c}{162.6\mu \times \left(\log\left(\frac{kt}{\phi\mu c_t r_c^{\ 2}} \right) - 3.23 \right)} \qquad \text{Eq. 8}$$

Translation: If a distance to a conduit has been entered, return the product of a conversion factor for injection rate, permeability, net thickness of the reservoir, and critical pressure, all divided by the product of 162.6, viscosity, and the difference of $\log_{10}$ of the quotient of permeability times the injection duration divided by the product of porosity, viscosity, total compressibility of the reservoir formation and reservoir liquid, and square of the distance to the suspected conduit, minus 3.23; otherwise, leave blank.

CALCULATIONS MADE IN THE NON-PRINTING AREA OF THE ZEI WORKSHEET

This table converts values in the units used to input data, such as gallons per minute to the units typically found in reservoir engineering equations, such as barrels per day.

	G	H	I
6 7	THESE ARE CONVERSIONS FROM FAMILIAR UNITS TO UNITS FOR EQUATIONS		
8 9	Injection rate, BPD	Duration, hours	Total volume, gals
10 11	Effective years of past i	total years of injection	total hours

Injection Rate, BPD

Cell G9 =C11/42*1440

Translation: Injection rate in BPD divided by 42 gal/bbl multiplied by 1440 min/day.

Duration, hrs

Cell H9 =C13*30.44*24+E13/C11/60

Translation: The product of the number of months of future injection times 30.44 days per month times 24 hours per day, plus the quotient of the current cumulative injected volume in gallons divided by both the absolute value of the injection rate in GPM and 60 minutes per hour.

Cumulative injected volume

Cell I9 =E13+C11*C13*30.44*1440

This is the total volume which is expected to be injected through the life of the well, including previously injected volume and the volume yet to be injected. It is calculated to estimate the radius of the waste plume:

$$V_{cum} = V_{past} + V_{fut} \qquad \text{Eq.9}$$

and:

$$V_{past} = t_{past} \times q \qquad \text{Eq. 10}$$

and:

$$V_{fut} = t_{fut} \times q \qquad \text{Eq. 11}$$

Where:

V_{cum} = the final cumulative volume including all past and projected future injection (may be entered directly or computed based on rate information and length of prior injection period);

V_{past} = volume already injected (E13);

V_{fut} = volume projected to be injected in the future (C11*C13*30.44*1440);

t_{past} = the time in minutes between date of first injection and today, calculated by subtracting the date number of the date of first injection from today's date number.

t_{fut} = the time in minutes for projected future injection, calculated by multiplying the number of months (C13) by the nominal number of days per month (30.44), times the number of minutes in a day (1440);

q = the injection rate in gallons per minute.

Translation: Past injection plus the absolute value of the average or permitted maximum injection rate times the duration of additional injection in months, times the number of days per average month times the number of minutes per day.

Effective years of past injection

Cell G11 =E13/(C11*1440*365.25)

Translation: Current cumulative injected volume divided by the projected rate in gallons per minute times 1440 minutes per day times 356.35 days per year.

Total years of injection

Cell H11 =IF(E11>"01/01/1901",TODAY()-DATEVALUE(E11)/365.25,C13/12)

Translation: If a value has been entered for date of first injection, return the result of dividing the period of future injection by 12 and adding the duration of past injection in years; otherwise, return the number of future months of injection divided by 12.

Total hours

Cell I11 =H11*365.25*24

Translation: Total effective years of injection times 365.25 days per year times 24 hours per day.

Calculation of the Exponential Integral

The calculation of the exponential integral is the most accurate means of calculating pressure increase. This table allows calculation of a complex mathematical function which is normally found through reference to tables. The calculations come from Attachment VI.A.1(a)-1 of the petition for exemption from the land disposal restrictions prepared by The Subsurface Group for submission by Environmental Disposal Systems, Inc.

	G	H	I	J
18 19 20	THESE ARE CALCULATIONS NEEDED TO CALCULATE THE CONE OF INFLUENCE USING THE EXPONENTIAL INTEGRAL METHOD			
21				
22	X	ln(1.781 x X)	el(-x)	
23				
24	G	A	B	INTERMED
25				

X, the argument for the exponential integral function

Cell G23 =948*C$21*D$30*(C$25+D$25)*D28^2/(C$23*H$9)

Translation: The constant 948 needed to account for unit conversions times porosity times viscosity in centipoise times the sum of the compressibility of the formation and the compressibility of the liquid in the pore spaces times the square of the radius of the distance in feet from the point of injection to the point at which pressure is being calculated divided by the product of permeability in millidarcies and the duration of injection in hours.

Log approximation of the exponential integral

Cell H23 =LN(1.781*G23)

This expression provides results which deviate very slightly from the exponential integral for arguments within the range of 0 to 0.02.

ei(-x)

Cell I23 =IF(G23<=0.02,-H23,IF(G23<=0.1,0-0.37528+0.95243*(-LN(G23)),IF(G23<=0.3,-0.099525+0.83515*(-LN(G23)),IF(G23<=1,0.21888+0.37839*(-LN(G23))+0.159733*(-LN(G23))^2,G25*(J25*EXP(-G23))))))

This formula partitions values of the Ei-function argument to correctly calculate the Ei value as shown in the table below by successively testing to find whether the value of the Ei-function argument is in the next higher range and then calculates the Ei value using the formulas shown in the table below.

Value of Equation 10 for distance r	Formula for exponential integral
<0.02	-H23
0.02 to 0.1	-0.37528 + 0.95243 × (-ln(Cell G23))
0.1 to 0.3	-0.099525 + 0.83515 × (-ln(Cell G23))
0.3 to 1.0	0.21888 + 0.37839 × (-ln (G23)) + 0.159733 × (-ln(G23))^2
>1.0	Cell G25 x J25 × exp(-Cell G23))

G, a factor in calculating the exponential integral

Cell G25 =1/G23

Translation: The inverse of the argument for the exponential integral function

A, a factor in calculating the exponential integral

Cell H25 =G25*(G25+3.37735)+2.05216

B, a factor in calculating the exponential integral

Cell I25 =G25*(1.072553*G25+5.716943)+6.945239

Intermediate step in the calculation of the exponential integral for very large values of X

Cell J25 =1-G25*(H25*G25+0.2709479)/(G25*(I25*G25+2.59388)+0.2709496)

Calculation of Viscosity

The table at right is based on linear fits to the graphical depiction of the relationship of viscosity to salinity and temperature. The reservoir temperature is provided to equations in column I and each equation is solved for viscosity as though the TDS content of the water were within the range indicated by the number just to the right and the number just below that. Equations in column G select the proper result found by the equations in Column I by testing the salinity value to see within which of the ranges described above the salinity falls.

	G	H	I
34	CALCULATION OF VISCOSITY		
35	Reservoir Temperature, F		0
36	0	260000	4.84
37	0	240000	3.91
38	0	200000	3.73
39	0	160000	2.98
40	0	120000	2.88
41	0	80000	2.41
42	0	40000	2.25
43	2.16	0	2.16

Reservoir Temperature

Cell I35 =Transfer!B59

Reservoir temperature is a factor in the determination of viscosity

Selection of viscosity

Cell G36 =IF(AND(Transfer!B$51<=H36,Transfer!B$51>H37),I36,0)

Salinity

Column H: Contains the upper concentration limits of the ranges in which the viscosities calculated in Column I are adequately accurate. See chart for cell contents.

Calculation of Viscosity

Column I - Rows 36 to 43

Cell I36 =IF(I35<80,1.64+(80-I35)*0.04,IF(I$35<120,1.1+(120-I$35)*0.0135,IF(I$35<160,0.77+(160-I$35)*0.00825, IF(I$35<200,0.6+(200-I$35)*0.00425,"Use chart"))))

Translation: If the temperature is less than 80° F, return 1.64 plus 0.04 times the difference between 80° F and the temperature; if the temperature is less than 120° F, return 1.1 plus 0.0135 times the difference between 120° F and the temperature; if the temperature is less than 160° F, return 0.77 plus 0.00825 times the difference between 160° F and the temperature; if the temperature is less than 200° F, return 0.6 plus 0.00425 to the difference between 200° F and the temperature; if none of these apply, display "Use chart".

Calculations in succeeding rows are similar, translated one row downward.

Check of calculation of ZEI

Purpose: This calculation uses a familiar equation and an alternate route for using base data to provide an indication if some step has become corrupted.

A	B	C	D	E
57	THIS SECTION IS USED TO CHECK THE CALCULATION OF RADIUS OF INFLUENCE BY			
58	CALCULATING THE PRESSURE BUILD UP AT THE EDGE OT THE CALCULATED CONE OF			
59	INFLUENCE. IF THE VALUE IN THE OUTLINED CELL EQUALS CALCULATED CRITICAL			
60	PRESSURE, CALCULATION IS CORRECT			
61				
62	The result of using the exponential integral solution must nearly match the value found in Cell C60			
63				

Cell D60 =-162.6*C11*42/1440*D30/(C23*B25)*(LOG10(C23*(C13*30.44*24+E13/abs(C11)/60)/(C21*D30*(C25+D25)*D28^2))-3.23)

Translation: The product of -162.6, the injection rate in gpm converted to bpd, the viscosity divided by the product of permeability and thickness, and the difference of the log of the permeability times all the hours of past and future injection divided by the product of porosity, viscosity, total compressibility, and the square of the ZEI radius (close log), minus 3.23.

Cell D63 =-70.6*C11*42/1440*D30/(C23*B25)*I23

Purpose: Another cross check using a different method to check the computation of the Exponential integral.

Translation: Product of negative 70.6, injection rate in gpm converted to bpd, and viscosity, divided by the product of effective thickness and permeability, all multiplied by the Exponential integral from I23.

FORMULAS USED ON PLOT WORKSHEET

Calculation of the linear portion of the equation for the exponential integral (Ei) solution of the diffusivity equation.

Cell I23 =70.6*'ZEI'!$G$9*'ZEI'!$D$30/('ZEI'!$C$23*'ZEI'!$B$25)
=70.6*q*effvis/(izhk*izh)

Purpose: This cell contains the solution of that part of the exponential integral method for calculating pressure increase which is outside the brackets.

Translation: The product of 70.6 (a constant accounting for units), the injection rate in barrels per day, and the viscosity of appropriate fluid in centipoise (cp) divided by the product of the permeability (md) and effective thickness (ft) of the injection zone.

Column I - Distance, ft

Purpose: To establish a series of increasing distances from the subject well at which to calculate pressure increase.

Cell I3 =ZEI!B13

Translation: The starting point for the transect will be at the formation face of the well.

Cell I4 = IF(I$3<0.005*I$22*J22,0.005*I$22*J22,I$3)

Translation: In this cell and in each lower row in this column, use the larger of the well radius or the calculated portion of the total ZEI distance. This prevents a graphing problem if the distance to the first point (Cell I3) is larger than the distance to a subsequently calculated point. Each cell in lower rows uses a set increase in the multiplier used to calculate distance relative to the radius of the ZEI. J22 is set at 2, but where another multiplier would result in a "more satisfying" view of the decline curve as a whole, a larger or smaller multiplier may be substituted in Cell J22.

Cell I15 =0.5*I22*J22

This row will contain the critical pressure. The plot continues to show an equivalent amount of decline beyond the limit of the ZEI.

Cell I21 =I22*J22

This is the last of the cells containing distances for the calculation of pressure.

Cell I22 =ZEI!$D28
=zei

The radius of the ZEI is D28 on the worksheet page. This distance is a factor in determining the length of the radial transect along which to show pressure decline.

Column J – Injection zone pressure extrapolated to base of the USDW

Calculations in this column estimate the pressures which would exist at the depth of the base of the lowermost USDW in a series of wells extending away from the center of injection if those wells are open to flow from the injection zone.

Cell J3-J21 ='ZEI'!E$21-I$23*R3-(+'ZEI'!B$23-'ZEI'!B$18)*'ZEI'!E$30*0.433
=ZEI!E$21-I$23*R3-(+ZEI!B$23-dusdw)*effsg*0.433

Purpose: Extrapolates the hydraulic pressure in the injection zone to the base of the USDW at a series of distances from the injection well. To the pre-existing pressure at the top of the injection zone, adds the pressure increase due to injection and decreases that result by the pressure lost between the top of the injection interval and the base of the USDW.

Translation: The pressure at the top of the injection zone minus the product of the solution of the portion of the Ei method outside the brackets and the value of the Ei function minus the product of the difference between the depths of the top of the injection zone (interval) and the depth to the base of the lowermost USDW, the specific gravity of the fluid in the formation at the distance of the limits of the ZEI, and the liquid pressure gradient of fresh water.

Cell J22: The value in Cell J22 is a multiplier which the analyst can adjust to show as much of a "tail" on the depiction of the pressure distribution as he believes is instructional. The default value is two (2).

Column K - Pressure at base of the lowermost USDW

This column calculates the pressure at the bottom of the lowermost USDW, providing the data foro the top plot and the "ZEI" page's plot to display this pressure as a solid, horizontal line.

Cells K3 through K21 =ZEI!D$18

Column L - Water Level, ft

Cell L3 ='ZEI'!B$11-'ZEI'!B$18+J3/(0.433*'ZEI'!E$30)
=gl+kb-dusdw+J3/(0.433*effsg)

This column converts the pressure due to injection at the depth of the base of the lowermost USDW to hydrostatic head measured in feet relative to sea level. Allows display of the piezometric surface on the diagrams.

$$H_{iz} = e_{surf} - d_{base} + \frac{p_{trans}}{0.433 psi/ft \times SG_{iz}} \quad \text{Eq. 12}$$

where:
H_{iz} =elevation of the piezometric surface along a transect away from the injection well;
e_{surf} = surface elevation of the wellhead,
d_{base} = the depth to the base of the lowermost USDW, in feet;
0.433 *psi/ft* = the liquid pressure gradient for fresh water
p_{trans} = pressure of the injection zone (interval) calculated at base of the USDW
SG_{iz} = the specific gravity of the water in the injection zone.

Translation: Calculate the depth to the base of the lowermost USDW with respect to mean sea level by subtracting the base of the USDW from the surface elevation of the well, then add the quotient of the injection zone (interval) pressure at the base of the USDW divided by the weight per foot of the injection zone fluid.

Column M - Surface Elevation, ft

Allows the display of reference elevation on the plot as a solid line.

Cells M3 through M20 ='ZEI'!B$11
=gl+kb

Translation: Display the reference elevation for the well head.

Column N - USDWS Base

Allows display of the elevation of the base of the lowermost USDW on Diagram #2.

Cells N3 through N20 ='ZEI'!B$11-'ZEI'!B$18
=gl+kb-dusdw

Translation: Reference elevation of the well minus the depth to the base of the lowermost USDW.

Column O - USDW Head, ft

This column calculates the hydrostatic head at the base of the lowermost USDW.

Cells O3 through O20 ='ZEI'!B$11-'ZEI'!B$18+'WorkSheet'!D$18/('ZEI'!D$16*0.433)
=gl+kb-dusdw+ZEI!D$18/(usdwsg*0.433)

Translation: Reference elevation of the wellhead minus the depth to the base of the lowermost USDW plus the quotient of the pressure at the base of the lowermost USDW divided by the weight of the fluid in the USDW per foot.

Column P -Ei-function argument

Calculates the Ei-function argument (value for which the exponential integral is found) of the solution for pressure increase. *(Smith)*

$$\left[-948\times\frac{\phi\times c_t\times\mu\times r^2}{k\times t}\right] \quad \text{Eq. 13}$$

Where:
948= Constant required to account for units;
N= average porosity of the effective portion of the injection zone;
c_t = total compressibility of the formation and contained fluid;
μ = viscosity of liquid dominating pressure buildup in the injection zone, cp;
r= radius at which the pressure increase is to be calculated;
k= permeability, md;
t= time, hours.

Cell P3 =948*''ZEI'!C$21*'ZEI'!D$30*('ZEI'!C$25+'ZEI'!D$25)*I3^2/('ZEI'!C$23*'ZEI'!H$9)
=948*izpor*effvis*(cl+cfm)*I3^2/(izhk*t)

Translation: The product of -948, porosity of the effective injection zone (interval), viscosity of the reservoir fluid beyond the limits of the ZEI, total compressibility of the injection reservoir rock and its contained fluid and the square of the distance to the point of calculation divided by the product of the permeability of the effective injection reservoir (interval) and the "effective" injection time.

Column Q - ln(1.781 x X)

Calculates the log approximation of the solution to the diffusivity equation for use when the Ei-function argument is less than 0.02. *(Smith, Eq. 2.33)*

Cell Q3 =LN(1.781*P3)

Translation: Calculate the natural log of the product of 1.781 and the Ei-function argument of the Ei equation for pressure increase.

Column R - ei(-x)

Chooses which calculation of the exponential integral to use in column J for the calculation of injection zone pressure translated to the elevation of the base of the lowermost USDW.

Cell R3 `=IF(P3<=0.02,-Q3,IF(P3<=0.1,0-0.37528+0.95243*(-LN(P3)),IF(P3<=0.3,-0.099525+0.83515*(-LN(P3)),IF(P3<=1,0.21888+0.37839*(-LN(P3))+0.159733*(-LN(P3))^2,S3*(V3*EXP(-P3))))))`

Translation: This formula partitions values of the Ei-function argument to correctly calculate the Ei value as shown in the table below by successively testing to find whether the value of the Ei-function argument is in the next higher range and then calculates the Ei value using the formulas shown in the table below.

Value of Equation 10 for distance r	Formula for exponential integral
<0.02	-ln(1.781 x Cell P3)
0.03 to 0.1	-0.37528 + 0.95243 x (-ln(Cell P3))
0.1 to 0.3	-0.099525 + 0.83515 x (-ln(Cell P3))
0.3 to 1.0	0.21888 + 0.37839 x (-ln(P3)) + 0.159733 x $(-\ln(P3))^2$
>1.0	Cell S3 x V3 x exp(-Cell P3))

Column S – Factor G in Column V

Calculates the inverse of the Ei-function argument needed to calculate the Ei function at very high values of the Ei-function argument.

Cell S3 `=1/P3`

Column T – Factor A in Column V

Calculates an early value needed to calculate the Ei function at very high values of the Ei-function argument.

Cell T3 `=S3*(S3+3.37735)+2.05216`

Column U – Factor B in Column V

Calculates an early value needed to calculate the Ei function at very high values of the Ei-function argument.

Cell U3 `=S3*(1.072553*S3+5.716943)+6.945239`

Column V – Intermediate

Calculates an intermediate value based on the results of calculations in columns S, T, and U. This value is needed to calculate the Ei function at high values of the Ei-function argument

Cell V3 =1-S3*(T3*S3+0.2709479)/(S3*(U3*S3+2.59388)+0.2709496)